
1 Introduction

Computational geometry is the design and analysis of algorithms for geometric problems that arise
on low dimensions, typically two or three dimensions. Many elegant algorithmic design and analysis
techniques have been devised to attack geometric problems. This is why I’ve included this topic in
this course.

Some applications of CG:

Computer Graphics

images creation

hidden surface removal

illumination

Robotics

motion planning

Geographic Information Systems

Height of mountains

vegetation

population

cities, roads, electric lines

CAD/CAM computer aided design/computer aided manufacturing

Computer chip design and simulations

Scientific Computation

Blood flow simulations

Molecular modeling and simulations

Basic algorithmic design approaches:

• Divide-and-Conquer

• Line-Sweep (typically in 2D)

• Random Incremental

In this course there will be three lectures on computational geometry covering the following topics:

• Geometric primitives

• Convex hull in 2D

• Random incremental algorithm for closest pair

• Divide and conquer algorithm for closest pair

• Sweep line algorithm for intersecting a set of segments

• An expected linear time algorithm for 2D linear programming.

1.1 Representations

The basic approach used by computers to handle complex geometric objects is to decompose the
object into a large number of very simple objects. Examples:

• An image might be a 2D array of dots.

• An integrated circuit is a planar triangulation.

• Mickey Mouse is a surface of triangles

It is traditional to discuss geometric algorithms assuming that computing can be done on ideal
objects, such as real valued points in the plane. The following chart gives some typical examples
of representations.

Abstract Object Representation

Real Number Floating Point Number, Big Number

Point Pair of Reals

Line Pair of Points, An Equation

Line Segment Pair of Endpoints

Triangle Triple of points

Etc

1.2 Using Points to Generate Objects

Suppose P1, P2, . . . Pk ∈ Rd. Below are several ways to use these points to generatel more complex
objects.

Linear Combination

Subspace =
∑

αiPi where αi ∈ R

Affine Combination

Plane =
∑

αiPi where
∑

αi = 1, αi ∈ R

Convex Combination

Body =
∑

αiPi where
∑

αi = 1, αi ≥ 0 αi ∈ R

2 Primitive Operations

I’ll be giving integer implementations of these primitives in ocaml. Let’s start with some basic
operations on vectors in 2D. The code below defines vector addition subtraction, cross product, dot
product and the sign of a number.

let (--) (x1,y1) (x2,y2) = (x1-x2, y1-y2)

let (++) (x1,y1) (x2,y2) = (x1+x2, y1+y2)

let cross (x1,y1) (x2,y2) = (x1*y2) - (y1*x2)

let dot (x1,y1) (x2,y2) = (x1*x2) + (y1*y2)

let sign x = compare x 0

(* returns -1 if x<0, 0 if x=0 and 1 if x>0 *)

Line Side Test

Given three points P1, P2, P3, the output of the line side test is “LEFT” if the point P3 is to the
left of ray P1 → P2, “RIGHT” if the point P3 is to the right of ray P1 → P2, and “ON” if it is on
that ray.

The algorithm is to construct vectors V2 and V3 by subtracting P1 from P2 and P3 respectively.
Then take the cross product of V2 and V2 and look at its value compared to 0.

Here is an implementation of this test in ocaml which returns 1 if p3 is LEFT of ray p1 → p2, −1
if RIGHT, and 0 if ON.

let line_side_test p1 p2 p3 = sign (cross (p2--p1) (p3--p1))

Line segment intersection testing

Here we are given two line segments (a, b) and (c, d) (where a, b, c, d are points), and we have to
determine if they cross. We can do this using four line-side tests, as illustrated here.

let segments_intersect (a,b) (c,d) =

(line_side_test a b d) * (line_side_test a b c) <= 0 &&

(line_side_test c d a) * (line_side_test c d b) <= 0

By changing the <= into a <, this can be changed into a strict intersection test, which would require
the segments to intersect at a point interior to both of them.

In-circle test

Three non-colinear points determine a circle. The in-circle test will tell us the relationship of a
fourth point to the circle determined by the other three points. So the test takes points a, b, c, and
d as inputs, and returns 1, 0, or −1 as follows:

This returns 0 if the four points are on the same circle (or straight line.) Suppose I
walk forward around the circle with my right hand on the circle from a → b → c. It
returns 1 if d is on the same side of the circle as my body, and −1 otherwise. It’s a
fourth degree function in the given coordinates.

let incircle (ax,ay) (bx,by) (cx,cy) (dx,dy) =

let det ((a,b,c),(d,e,f),(g,h,i)) =

a*(e*i - f*h) - b*(d*i - f*g) + c*(d*h - e*g)

in

let row ax dx ay dy =

let a = ax - dx in

let b = ay - dy in

(a, b, (a*a) + (b*b))

in

sign (det (row ax dx ay dy, row bx dx by dy, row cx dx cy dy))

Many computational geometry applications use numerical tests known as the orientation and incircle tests. The orientation test determines
whether a point lies to the left of, to the right of, or on a line or plane defined by other points. The incircle test determines whether a point
lies inside, outside, or on a circle defined by other points.

Each of these tests is performed by evaluating the sign of a determinant (see the figure below). The determinant is expressed in terms of the
coordinates of the points. If these coordinates are expressed as single or double precision floating-point numbers, roundoff error may lead
to an incorrect result when the true determinant is near zero. In turn, this misinformation can lead an application to fail or produce incorrect
results.

One way to solve this problem is to use exact arithmetic. Unfortunately, traditional libraries for arbitrary precision floating-point arithmetic
are quite slow, and can reduce the speed of an application by one or two orders of magnitude.

To minimize this problem, I've produced algorithms and implementations for performing the 2D and 3D orientation and incircle tests

The picture above illustrates a case when the incircle test would return 1. (This figure was taken
from http://www.cs.cmu.edu/~quake/robust.html)

3 Computing the Convex Hull

This is the “sorting problem” of computational geometry. There are many algorithms for the
problem, and there are often analogous to well-known sorting algorithms.

A point set A ⊆ Rd is convex if it is closed under convex combinations. That is, if we take any
convex combination of any two points in A, the result is a point in A.

Definition: ConvexClosure(A) = CC(A) = smallest convex set containing A

There are two different definitions of the convex hull. The first one is the one we will use. The
second is used by some books.

Definition 1: ConvexHull(A) = boundary of CC(A)

Definition 2: ConvexHull(A) = CC(A)

A computer representation of a convex hull must include the combinatorial structure. In two
dimensions, this just means a simple polygon in, say clockwise order. (In three dimensions it’s a
planar graph of vertices and edges. The vertices are a subset of the input points.)

3.1 Graham Scan for 2D convex hulls

The input is a set of points {P1, P2, . . . Pn}. Find the point with the smallest y coordinate and
swap it with P1. Now for each point Pi, compute the angle of the ray P1 → Pi, and sort the points
by this angle. (These angles are all between 0 and π.) (Actually in the code below there is no use
of floating point numbers. The sorting is done by the use of line side tests.)

Start with segment (P1, P2). In general there will be a sequence (or “chain” as it’s called in the
code below) of segments, starting from (P1, P2 · · · ). The sequence is “turning left” which means
that each time we move from one segment to the next we make a left turn. Our goal is to process
a new point, and add it onto this sequence.

Here’s the algorithm. Add the new point Pi to the end of this sequence. If the sequence of segments
now has a right turn at the end, then delete the second to last point of this sequence. Repeat the
process of deleting the second to last element until the sequence is restored to one that has only
left turns. The direction of a turn can be determined using a line side test.

After all the points are processed in this way, we can just add the last segment from Pn to P1, to
close the polygon, which will be the convex hull.

Each point can be added at most once to the sequence of vertices, and each point can be removed
at most once. Thus the running time of the scan is O(n). But remember we already paid O(n log n)
for sorting the points at the beginning of the algorithm, which makes the overall running time of
the algorithm O(n log n).

(Alternatively, keep a token on each segment of the current sequence. Use the token to pay for a
segment’s removal. Adding a new segment is one unit of work, plus another token to keep on it.)

Complete ocaml code for the graham scan is at the end of these notes.

This figure shows the progress of the algorithm. The points are numbered in order of increasing
angle. The red line shows the current sequence of segments after some points have been processed.
Note that when point 5 is added point 4 is removed from the sequence, and when point 7 is added,
points 6 and 5 are both removed.

3.2 Lower bound for computing the convex hull

Suppose the input to a sorting problem is X1, . . . , Xn. Consider computing the convex hull of the
following set of points:

(X1, X
2
1 ), . . . , (Xn, X

2
n)

All of these points are on the convex hull (they’re on a parabola). Thus they are returned in the
order they appear along the parabola. No matter which convex hull algorithm is used, the points
can be reflected and/or cyclically shifted so that their x coordinates are in sorted order. Thus, they
can be sorted by computing a convex hull followed by O(n) additional work. Thus any comparison
based convex hull algorithm must make Ω(n log n) comparisons. The figure below illustrates this
phenomenon.

3.3 Ocaml code for the Graham Scan convex hull algorithm

let (--) (x1,y1) (x2,y2) = (x1-x2, y1-y2)

let (++) (x1,y1) (x2,y2) = (x1+x2, y1+y2)

let cross (x1,y1) (x2,y2) = (x1*y2) - (y1*x2)

let dot (x1,y1) (x2,y2) = (x1*x2) + (y1*y2)

let sign x = compare x 0

let line_side_test p1 p2 p3 = sign (cross (p2--p1) (p3--p1))

(* Which side of ray p1-->p2 is p3 on? This returns 1 for

"LEFT", 0 for "ON" and -1 for "RIGHT". *)

let len (x,y) =

let sq a = a*.a in

let (x,y) = (float x, float y) in

sqrt ((sq x) +. (sq y))

let graham_convex_hull points =

let inf = max_int in

let base = List.fold_left min (inf,inf) points in

let points = List.sort (

fun pi pj ->

if pi=pj then 0

else if pi=base then 1

else if pj=base then -1

else line_side_test base pj pi

) points in

(* now the list starts at p1, and base is at the end of the list *)

let rec scan chain points =

let (c1,c2,chainx) = match chain with

| c1::((c2::_) as chainx) -> (c1,c2,chainx)

| _ -> failwith "chain must have length at least 2"

in

match points with [] -> chain

| pt::tail ->

match line_side_test c2 c1 pt with

| 1 -> scan (pt::chain) tail

| -1 -> scan chainx points

| _ ->

if len (pt--c2) > len (c1--c2)

then scan (pt::chainx) tail

else scan chain tail

in

match points with

| (p1::((_::(_::_)) as rest)) -> List.tl(scan [p1;base] rest);

| _ -> points (* do nothing if < 3 points *)

let print_list l =

List.iter (fun (x,y) -> Printf.printf "(%d,%d) " x y) l;

print_newline()

let () = print_list (graham_convex_hull [(0,0);(0,2);(2,2);(2,0);(1,1)])

Running the algorithm:

$ ocamlc graham.ml

$ ./a.out

(0,2) (2,2) (2,0) (0,0)

